

IE360: CAD/CAM

Computer Aided Design and Computer
Aided Manufacturing

Lecture (12)

Computer-Assisted Part Programming

Dr Ashraf Afifi

Industrial Engineering Department
King Saud University

Outline

- Automatically Programmed Tool (APT) Language
- Syntax of APT
 - Geometry statements
 - Motion statements
 - Additional APT statements
 - Other capabilities of APT
- Examples
- Exercises

APT Language :

- A great number of programming languages have been developed for NC programming, of which the **Automatically Programmed Tool (APT)** language is the most comprehensive and widely used.
- The first prototype of the APT system was developed at MIT in 1956.
- The program was developed further by the cooperative efforts of many companies.
- As a result of these efforts, APT II was developed in 1958, and a more effective system, APT III, was made available in 1961.
- The present APT language can control machines with as many as five motion axes.

Syntax of APT:

➤ An APT program is composed of statements that belong to one of five types:

- *Identification statements*, which specify part name and the specific post-processor.
- *Geometry statements*, which define the part geometry relevant to the machining operations,
- *Motion statements*, which define the motion of the cutting tool with respect to the part geometry,
- *Post-processor statements*, which specify machining parameters such as feed, speed and coolant on/off.
- *Auxiliary statements*, which specify auxiliary machine-tool functions to specify the tool, tolerance, and the like.

Geometry statements:

- The general form of a geometry statement is as follows:

symbol = geometry_word/descriptive data

- The *symbol* is a name for the geometric element and has the same role as a variable name in other high level languages.
- The *geometry_word* specifies the type of geometry, such as points, lines, planes and circles.
- The *descriptive data* are the numeric data required to define the entity.

Geometry statements:

➤ Example:

P1 = POINT/3.0, 2.0, -1.0

P2 = POINT/6.0, 5.0, 3.0

L1 = LINE/P1, P2

PL1 = PLANE/P1, P2, P3

C1 = CIRCLE/CENTER, P1, RADIUS, 1.5

- The first two statements define two points P_1 and P_2 with coordinates (3, 2, -1) and (6, 5, 3), respectively.
- The third statement creates a line L_1 between the predefined points P_1 and P_2 .
- The fourth statement defines a plane with three points P_1 , P_2 , and P_3 that are not on the same straight line.
- The last statement creates a circle with P_1 as a center and a radius of 1.5.
- Other ways exist to define the above geometries.

Motion statements:

➤ Two groups of motion statements are available: one for **point-to-point machining** and the other for **continuous-path (contouring) machining**.

➤ Syntax of point-to-point machining:

FROM/*point_location*

GOTO/*point_location*

GODLT/ $\Delta x, \Delta y, \Delta z$

▪ The *point_location* may be given in terms of the x , y , and z coordinates, or it may be a symbolic of the point that has been previously defined in a geometry statement.

▪ The FROM statement specifies the initial location from which a motion starts.

▪ The GOTO statement will move the tool rapidly along a straight line from the tool's present location to the point specified by this statement.

- The GODLTA statement will move the tool the incremental distance specified by Δx , Δy , Δz from its present location.

➤ Example:

```
FROM/P1  
GOTO/3.0, 4.0, -2.0  
GODLTA/2.0, -1.0, 0.0
```

- The first statement instructs the tool to move from its current position specified by point P_1 .
- The second statement will move the tool from its current position to the point specified by $x = 3.0$, $y = 4.0$, and $z = -2.0$.
- The third statement will move the tool from its current position to a new position defined by $\Delta x = 2.0$, $\Delta y = -1.0$, and $\Delta z = 0.0$.

➤ Syntax of continuous-path (contouring) machining: A motion statement for continuous-path machining requires three control surfaces to guide the tool motion: *part*, *drive* and *check* surfaces as shown in Figure 1:

- The *part surface* is a surface on which the end of the tool is riding.
- The *drive surface* is a surface along which the tool slides.
- The *check surface* is a surface bounding the tool motion so that the tool motion continues until the check surface is encountered.




Figure 1: Tool guiding surfaces in APT continuous-path machining

➤ In APT, the cutting tool moves along the control surfaces using a GO command as shown in the following syntax:

GO/TO, drive surface, TO, part surface, TO, check surface

- The “TO” in the above statement is a modifier. Other modifiers are “ON”, and “PAST”.

- These modifiers indicate the desired location of the cutter with respect to the associated control surface. Figure 2 shows these meanings graphically.




Figure 2: Meaning of TO, ON, and PAST modifiers in APT

➤ The GO command considered how to locate the tool at its initial position with respect to the three control surfaces. The following commands consider moving the tool relative to the previous motion direction:

GOLFT/
GORGT/
GOUP/
GODOWN/

- The “GOLFT” command will move the tool left from the previous direction and along the drive surface.
- “GORGT” will move the tool right from the previous direction and along the drive surface.
- “GOUP” will move the tool up along the drive surface (i.e., away from the part surface).
- “GODOWN” will move the tool down along the drive surface (i.e., closer to the part surface).

Additional APT statements:

➤ Geometry and motion statements represent about two-thirds of an average APT program. The statements composing the remainder of the program can be classified as post-processor statements, tolerance and cutter statements and initial and termination statements. Sample commands are:

```
MACHIN/  
COOLNT/  
FEDRAT/  
SPINDL/  
TOOLNO/  
CUTTER/  
END  
FINI
```

- The “MACHIN/” statement is used to specify the machine tool, for example, MACHIN/DRILL, 2 might specify the second NC drill in the shop.


- The “COOLNT/” statement allows the coolant fluid to be turned on and off as in: COOLNT/ON and COOLNT/OFF.
- The “FEDRAT/” statement specifies the feed rate for moving the tool – for example, FEDRAT/4.5 in inches per minute.
- The “SPINDL/” statement turns on the spindle or specifies the spindle rotation speed in revolution per minute, as in: SPINDL/ON and SPINDL/1250, CCLW.
- The “TOOLNO/” statement specifies the tool number to be used – for example, TOOLNO/3572, 6 specifies tool 3572 of 6 unit length.
- The “CUTTER/” statement is used to specify the cutter diameter – for example, the statement CUTTER/0.6 is for a tool of diameter 0.6 unit.
- The “END” statement forces a machine tool to stop so that the operator can manually perform an inspection or perhaps change a tool.
- The “FINI” statement specifies the end of the program.

Other capabilities of APT:


- Similar to the conventional high-level programming languages, APT also provides facilities for arithmetic manipulation and looping and a subprogram feature known as the *macro facility*.
- This feature allows programming of repetitive operations as a subprogram and to call it repeatedly within a program.
- The parameters used in the subprogram can have variable names so that any values of them can be assigned when the macro is called. This is the same as the use of variable arguments in a FORTRAN subroutine.

Examples:


- **Example 1:** Write an APT program to drill two 0.75-inch diameter holes in a 1.0-inch thickness plate, as shown in Figure 3a. Use the tool home position shown, a drill speed of 500 rpm and a feedrate of 3.55 ipm. The machine that performs the drilling is number 5 and its controller is coded as DRILL. The tool path with the order in which the holes are drilled is shown in Figure 3b.
- **Example 2:** Rewrite the part program in Example 1 using the macro facility.
- **Example 3:** Write an APT program to mill the contour of the part shown in Figure 3a. Use an end-mill cutter with a diameter of 0.75 inch, a cutting speed of 580 rpm and a feedrate of 2.3 ipm. Milling machine name is MILL5. Figure 3c shows the tool path assuming a home position at P0 (0, -1.5, 0).



(a) Part geometry



(b) Drilling tool path



(c) Milling (contouring) tool path

Figure 3: Machining a typical part

Answers:

➤ Example 1:

```
PARTNO PTP-EXAMPLE  
MACHIN/DRILL, 5  
CUTTER/0.75  
P0 = POINT/-2.0, 0.0, 0.0  
P1 = POINT/1.25, 1.0, 0.0  
P2 = POINT/4.75, 1.0, 0.0  
SPINDL/500  
FEDRAT/3.55  
COOLNT/ON  
FROM/P0  
GOTO/P1  
GODLTA/0, 0, -1.5  
GODLTA/0, 0, 1.5  
GOTO/P2  
GODLTA/0, 0, -1.5  
GODLTA/0, 0, 1.5  
GOTO/P0  
COOLNT/OFF  
END  
FINI
```

Answers:

➤ Example 2:

PARTNO PTP-EXAMPLE

MACHIN/DRILL, 5

CUTTER/0.75

P0 = POINT/-2.0, 0.0, 0.0

SPINDL/500

FEDRAT/3.55

COOLNT/ON

FROM/P0

CALL/DRILL, X = 1.25, Y = 1.0, Z = 0.0, DEPTH = 1.5

CALL/DRILL, X = 4.75, Y = 1.0, Z = 0.0, DEPTH = 1.5

GOTO/P0

COOLNT/OFF

END

FINI

DRILL = MACRO/X, Y, Z, DEPTH

GOTO/X, Y, Z

GODLTA/0, 0, -DEPTH

GODLTA/0, 0, DEPTH

TARMAC

➤ Example 3:

PARTNO MILLING-EXAMPLE

MACHIN/MILL, 5

CUTTER/0.75

P0 = POINT/0, -1.5, 0

P1 = POINT/0, 0, 0

P2 = POINT/6.0, 0, 0

P3 = POINT/6.0, 2.0, 0

P4 = POINT/4.0, 2.0, 0

P5 = POINT/4.0, 4.0, 0

P6 = POINT/2.0, 4.0, 0

P7 = POINT/2.0, 2.0, 0

P8 = POINT/0, 2.0, 0

L1 = LINE/P1, P2

L2 = LINE/P2, P3

L3 = LINE/P3, P4

L4 = LINE/P4, P5

L5 = LINE/P5, P6

L6 = LINE/P6, P7

L7 = LINE/P7, P8

L8 = LINE/P8, P1

PL1 = PLANE/0, 0, -1.5, 6.0, 0, -1.5, 0, 2.0, -1.5

SPINDL/580

FEDRAT/2.30

COOLNT/ON

FROM/P0

GO/TO, L1, TO, PL1, TO, L8

GORGTL/L1, PAST, L2

GOLFT/L2, PAST, L3

GOLFT/L3, TO, L4

GORGTL/L4, PAST, L5

GOLFT/L5, PAST, L6

GOLFT/L6, TO, L7

GORGTL/L7, PAST, L8

GOLFT/L8, PAST, L1

GOTO/P0

COOLNT/OFF

END

FINI

Exercise 5:

➤ From the following APT source program, sketch the geometry and the tool path. Also, explain the meaning of the post-processor statements.

```
PARTNO PART3
MACHIN/MILL, 1
CUTTER/0.5
P0 = POINT/-4, -4, 0
P1 = POINT/0, 0, 0
P2 = POINT/-2, 4, 0
P3 = POINT/5, 4, 0
P4 = POINT/5, 0, 0
L1 = LINE/P1, P2
L2 = LINE/P2, P3
L3 = LINE/P3, P4
L4 = LINE/P4, P1
PL1 = PLANE/P1, P2, P3
```

....

```
....
SPINDL/1000
FEDRAT/5
COOLANT/ON
FROM/P0
GO/TO, L1, TO, PL1, TO, L4
GOLFT/L1, PAST, L2
GORGT/L2, PAST, L3
GORGT/L3, PAST, L4
GORGT/L4, PAST, L1
GOTO/P0
COOLANT/ OFF
FINI
```

Exercise 6:

➤ Write an APT program using the macro facility to drill two 0.75-inch diameter holes in a 1.5-inch thickness plate, as shown in the figure that follows. Use the tool home position shown, a drill speed of 500 rpm and a feedrate of 3.55 ipm. The machine that performs the drilling is number 5 and its controller is coded as DRILL. Also, write an APT program to mill the contour of the part shown. Use an end-mill cutter with a diameter of 0.75 inch, a cutting speed of 580 rpm and a feedrate of 2.3 ipm. Milling machine name is MILL5.

